
Generic Programming with Multiple Parameters

José Pedro Magalhães

Abstract. Generic programming, a form of abstraction in programming lan-
guages that serves to reduce code duplication by exploiting the regular structure
of algebraic datatypes, has been present in the Haskell language in different forms
for many years. Lately, a library for generic deriving has been given native sup-
port in the compiler, allowing programmers to write functions such as fmap that
abstract over one datatype parameter generically. The power of this approach is
limited to dealing with one parameter per datatype, however. In this paper, we
lift this restriction by providing a generalisation of generic deriving that supports
multiple parameters, making essential use of datatype promotion and kind poly-
morphism. We show example encodings of datatypes, and how to define a map
function that operates on multiple parameters simultaneously.

1 Introduction

Haskell [12], a pure, lazy, strongly-typed functional programming language, has been
a research vehicle for generic programming almost since its inception [2]. Generic pro-
gramming is a form of abstraction which exploits the structure of algebraic datatypes
in order to define functionality that operates on many datatypes uniformely, reducing
code duplication. With generic programming, certain functions (such as data seriali-
sation and traversals) can be written once and for all, working for existing and future
types of data.

Early approaches to generic programming in Haskell were separate extensions to
the compiler or preprocessors [3], but recently all approaches are bundled as libraries
[13], sometimes with direct compiler support [11]. The easier availability of generic
programming functionality appears to increase its usage; in particular, offering native
support for an approach in the compiler confers a sense of stability to a specific ap-
proach, allowing more programmers to use and benefit from generic programming.

This switch to library approaches simplifies the task of the generic programming
library developer, since it is typically easier to maintain a library than a separate ex-
tension. It also presents the library user with fewer barriers to adoption, as it requires
no external tools. However, it can compromise expressivity and usability, as library
approaches are limited to the capabilities of the Haskell language itself. Performance
and quality of error messages are common complaints of generic programming li-
braries, but also reduced support for certain datatypes can be a concern. In particular,
the Generic Haskell compiler [6], an early, pre-processor approach, had full support for
generic functions abstracting over multiple parameters. The same goes for the gener-
ics in the Clean language [1], which were implemented in the kind-indexed style of

Generic Haskell. However, to our knowledge no library approach to generic program-
ming in Haskell has native support for abstraction over multiple datatype parameters.
Concretely, consider the following example:

data WTree α ω = Leaf α

| Fork (WTree α ω) (WTree α ω)
| WithWeight (WTree α ω) ω

mapWTree :: (α → α ′)→ (ω → ω ′)→WTree α ω →WTree α ′ ω ′

mapWTree f g (Leaf a) = Leaf (f a)
mapWTree f g (Fork l r) = Fork (mapWTree f g l) (mapWTree f g r)
mapWTree f g (WithWeight t w) = WithWeight (mapWTree f g t) (g w)

Clearly, functions like mapWTree follow the structure of the datatype, and should be
defined generically, once and for all. In this paper we focus on the problem of defininig
functions such as mapWTree generically; many (if not all) modern generic programming
approaches support the WTree datatype, but not generic operations such as mapWTree.
The only exception that we are aware of is a convoluted implementation using Scrap
Your Boilerplate [5], which relies on runtime type comparison, casting, and (virtual)
seralisation.1 Our approach focuses instead on encoding the parameters adequately in
the generic representation.

In this paper, we focus our attention on the generic-deriving approach, as im-
plemented in the Glasgow Haskell Compiler (GHC) [8, Chapter 11]. To balance com-
plexity and ease-of-use, generic-deriving was designed from the start with one com-
promise in mind: many generic functions (e.g. fmap and traverse) abstract over one
datatype parameter (i.e. they operate on type containers, or types of kind ?→ ?), but
few require abstraction over more than one parameter. As such, the design of gener-
ic-deriving could be kept relatively simple, at the loss of some generality. Recently,
with the advent of kind polymorphism in GHC [16], many libraries are being gen-
eralised from kind-specific to kind-polymophic (e.g. Typeable [16]). In this paper we
describe an elegant generalisation of generic-deriving that works with multiple pa-
rameters, lifting the one-parameter restriction without requiring the full-blown power
(and complexity) of indexed functors [7]. Our solution is based on an earlier, failed
attempt [9, Section 6.1], but uses new insights to overcome old challenges: we use het-
erogeneous collections [4] to encode a function with multiple arguments as a regular
single-argument function that takes a list of arguments.

In the remainder of this paper, we first review one-parameter generic-deriving
(Section 2). We proceed to describe our generalisation in Section 3, showing example
datatype encodings (Section 3.2), and a generalised map function (Section 3.3). We list
limitations and propose future work in Section 4, and conclude in Section 5.

Notation In order to avoid syntactic clutter and to help the reader, we adopt a liberal
Haskell notation in this paper. Datatype promotion makes datatype definitions also de-
fine a kind (with the name of the datatype being defined), while the constructors also be-
come the types that inhabit that kind; we assume the existence of a kind keyword, which
allows us to define kinds that behave as if they had arisen from datatype promotion, ex-

1 http://okmij.org/ftp/Haskell/generics.html#gmap

http://okmij.org/ftp/Haskell/generics.html#gmap

cept that they do not define a datatype and constructors. This helps us preventing name
clashes. We omit the keywords type family and type instance entirely, making type-
level functions look like their value-level counterparts. We colour constructors in blue,
types in red, and kinds in green. Promoted lists are prefixed with a quote, to distinguish
them from the list type. Additionally, we use Greek letters for type variables, apart from
κ , which is reserved for kind variables. A grayscale version of this paper is available
at http://dreixel.net/research/pdf/gpmp_nocolour.pdf. The syntactic sugar is
only for presentation purposes. An executable version of the code, which compiles with
GHC 7.6.2, is available at http://dreixel.net/research/code/gpmp.zip.

2 Generic programming with one parameter

In this section we review the generic-deriving library (in the slightly revised version
of Magalhães and Löh [10]), which supports abstraction over (at most) one datatype
parameter. We omit meta-information, as it is not relevant to our development.

2.1 Universe

Using datatype promotion and kind polymorphism, we can keep the “realm” of generic
representations of user datatypes separate from the realm of user datatypes by defin-
ing a new kind. Like types classify values, kinds classify types. User datatypes always
have kind ?. For the generic representation, we define a kind Univ that aggregates the
types used to represent user datatypes in generic-deriving. U encodes constructors
without arguments, P an occurence of the parameter, K a type that does not contain
the parameter, R a type that contains the parameter, (:+:) the choice between two con-
structors, (:×:) is used for constructors with multiple arguments, and (:@:) encodes the
application of a functor to a type. The interpretation datatype In υ ρ encodes the values
associated with a representation type υ and parameter ρ:

kind Univ =
U
| P
| K ?
| R (?→ ?)
| Univ :+: Univ

| Univ :×: Univ
| ?→ ? :@: Univ

data In (υ :: Univ) (ρ ::?) ::? where
U1 :: In U ρ

Par1 :: ρ → In P ρ

K1 :: α → In (K α) ρ

Rec1 :: φ ρ → In (R φ) ρ

L1 :: In φ ρ → In (φ :+: ψ) ρ

R1 :: In ψ ρ → In (φ :+: ψ) ρ

(:×:) :: In φ ρ → In ψ ρ → In (φ :×: ψ) ρ

App1 :: φ (In ψ ρ) → In (φ :@: ψ) ρ

There is some redundacy in the universe codes; for example, R φ is essentially a shortcut
for φ :@: P, but the latter has a more complicated representation (φ (In P ρ) instead of
just φ ρ). Our new encoding in Section 3 also solves this issue.

The Generic type class mediates between user datatypes and their generic repre-
sentation. A Rep type function is used to encode the generic representation of some
user datatype α , while Par identifies the parameter. Conversion functions from and to
witness the isomorphism between α and In (Rep α) (Par α):

http://dreixel.net/research/pdf/gpmp_nocolour.pdf
http://dreixel.net/research/code/gpmp.zip

class Generic (α ::?) where
Rep α :: Univ
Par α ::?
from :: α → In (Rep α) (Par α)
to :: In (Rep α) (Par α)→ α

Instances of this class are trivial, but tedious to write, and are necessary for each datatype
intended to be used generically. Fortunately, these instances are automatically derivable
by GHC, therefore making generic-deriving a “built-in” generic programming ap-
proach.

2.2 Datatype encodings

The universe of generic-deriving is best understood through sample datatype instan-
tiations, which we provide in this section. We start with the encoding of lists. A list
is a choice ((:+:)) between an empty constructor (U) or two ((:×:)) arguments—the
parameter (P) and another list (R [ ]):

instance Generic [α ] where
Rep [α ] = U :+: P :×: R [ ]
Par [α ] = α

from [ ] = L1 U1
from (h : t) = R1 (Par1 h :×: Rec1 t)
to (L1 U1) = [ ]
to (R1 (Par1 h :×: Rec1 t)) = h : t

From this point forward we shall omit the to function, as it is always entirely symmetric
to from.

A slightly more complicated encoding is that of rose (or multiway) trees. Since we
use application of lists in the representation, the conversion functions need to be able to
map over these lists. This is achieved using the fmap function, a generic map that can
be defined for all Generic types (Section 2.3):

data RTree α = RTree α [RTree α ]

instance Generic (RTree α) where
Rep (RTree α) = P :×: ([ ] :@: R RTree)
Par (RTree α) = α

from (RTree x xs) = Par1 x :×: App1 (fmap Rec1 xs)

2.3 Mapping

As generic-deriving supports abstraction over one parameter, we can define the stan-
dard function fmap generically. The user-facing class for this function abstracts over
type containers φ of kind ?→ ?:

class Functor (φ ::?→ ?) where
fmap :: (α → β )→ φ α → φ β

The generic version, fmapR, on the other hand, operates on the generic representation.
We need to give an instance of fmapR for each representation type; we use another type
class for this purpose:

class FunctorR (υ :: Univ) where
fmapR :: (α → β )→ In υ α → In υ β

Sums, products, units, and constants are trivial:

instance FunctorR U where
fmapR U1 = U1

instance FunctorR (K α) where
fmapR (K1 x) = K1 x

instance (FunctorR φ ,FunctorR ψ)⇒ FunctorR (φ :+: ψ) where
fmapR f (L1 x) = L1 (fmapR f x)
fmapR f (R1 x) = R1 (fmapR f x)

instance (FunctorR φ ,FunctorR ψ)⇒ FunctorR (φ :×: ψ) where
fmapR f (x :×: y) = fmapR f x :×: fmapR f y

More interesting are the instances for: the parameter, where we apply the mapping func-
tion; recursion into type containers, where we recurse using fmap; and application of
type containers, where we fmap on the outer container, and fmapR on the inner repre-
sentation:

instance FunctorR P where
fmapR f (Par1 x) = Par1 (f x)

instance (Functor φ)⇒ FunctorR (R φ) where
fmapR f (Rec1 x) = Rec1 (fmap f x)

instance (Functor φ ,FunctorR υ)⇒ FunctorR (φ :@: υ) where
fmapR f (App1 x) = App1 (fmap (fmapR f ) x)

Providing instances for types with a Generic instance, such as [α ], requires only using
fmapR and converting from/to the original datatype:

instance Functor [ ] where
fmap f = to◦ fmapR f ◦ from

We now have an easy way to define a generic fmap for any user datatype with a Generic
instance.

3 Generic programming with multiple parameters

Having seen the current implementation of generic-deriving, we are ready to explore
the changes that are necessary to make it support abstraction over multiple parameters.

3.1 Universe

The main change to the universe involves the introduction of a list of parameters, and
the separation of fields into a kind of their own. A field can be one of three things.

Constant types (K) are unchanged. Parameters (P ν) now take an argument ν :: Nat
to indicate which of the parameters it is. The kind Nat encodes Peano-style natural
numbers; we will write 0 for Ze, 1 for Su Ze, etc. A generalised form of application
((:@:)) replaces R and the old (:@:), encoding the application of a type of kind κ to a
type-level list of fields. As we only deal with parameters of kind ? (see Section 4.1), the
first argument to :@: will always have a kind of the form ?→ . . .→ ?, and the second
argument will have as many elements as necessary to fully saturate the first argument.
The interpretation now takes a list of parameters ρ instead of a single parameter. A
separate InField υ ρ datatype interprets a field representation type υ with parameters ρ .
Constants are interpreted as before. Parameters are looked-up in the parameter list with
a type-level lookup operator (:!:) akin to the value-level (!) operator. An application
σ :@: χ is interpreted by applying σ to each of the arguments in χ , after transforming
these (with ExpandField) into types of kind ?:

kind Univ =
U
| F Field
| Univ :+: Univ

| Univ :×: Univ
kind Field =

K ?
| P Nat
| ∀κ.κ :@: [Field ]

data In (υ :: Univ) (ρ :: [? ]) ::? where
U :: In U ρ

F :: InField υ ρ → In (F υ) ρ

L :: In α ρ → In (α :+: β ) ρ

R :: In β ρ → In (α :+: β ) ρ

:×: :: In α ρ → In β ρ → In (α :×: β ) ρ

data InField (υ :: Field) (ρ :: [? ]) ::? where
K :: α → InField (K α) ρ

P :: ρ :!: ν → InField (P ν) ρ

A :: AppFields σ χ ρ → InField (σ :@: χ) ρ

kind Nat = Ze | Su Nat
(ρ :: [?]) :!: (ν :: Nat) ::?
(α ‘: ρ) :!: Ze = α

(α ‘: ρ) :!: (Su ν) = ρ :!: ν

AppFields σ χ ρ = σ :$: ExpandField χ ρ

(σ :: κ) :$: (ρ :: [? ]) ::?
σ :$: ‘[ ] = σ

σ :$: (α ‘: β ) = (σ α) :$: β

ExpandField (χ :: [Field ]) (ρ :: [? ]) :: [? ]
ExpandField ‘[ ] ρ = ‘[ ]
ExpandField ((K α) ‘: χ) ρ = α ‘: ExpandField χ ρ

ExpandField ((P ν) ‘: χ) ρ = (ρ :!: ν) ‘: ExpandField χ ρ

ExpandField ((σ :@: ω) ‘: χ) ρ = (σ :$: ExpandField ω ρ) ‘: ExpandField χ ρ

The ExpandField type function converts a list of Fields into user-defined types (of
kind ?). Constants are represented by the type in question, parameters are looked up
in the parameter list, and applications are expanded into a fully applied type.

The Generic class to mediate between user datatypes and their representation now
has a type function Pars which lists the parameters of the datatype:

class Generic (α ::?) where
Rep α :: Univ
Pars α :: [? ]
from :: α → In (Rep α) (Pars α)
to :: In (Rep α) (Pars α)→ α

3.2 Datatype encodings

To better understand the universe of our generalised generic-deriving, we show sev-
eral example datatype encodings in this section. We begin with lists; their encoding is
similar to that of Section 2.2, only now we use (:@:) instead of R:

instance Generic [α ] where
Rep [α ] = U :+: F (P 0) :×: F ([ ] :@: ‘[P 0])
Pars [α ] = ‘[α]

from [ ] = L U
from (h : t) = R (F (P h) :×: F (A t))

The RTree type of Section 2.2 can also still be encoded. In fact, since we use the
type family ExpandField to transform fields υ of kind Field into ?, instead of using the
interpretation InField υ recursively, we no longer need to use fmap; the arguments can
be used directly, simplifying the implementation of from and to:

instance Generic (RTree α) where
Rep (RTree α) = F (P 0) :×: F ([ ] :@: ‘[RTree :@: ‘[P 0]])
Pars (RTree α) = ‘[α]

from (RTree x xs) = F (P x) :×: F (A xs)

Having support for multiple parameters, we can now deal with pairs properly, for
example. These are simply a product between two fields with a parameter each:

instance Generic (α,β ) where
Rep (α,β ) = F (P 0) :×: F (P 1)
Pars (α,β ) = ‘[α,β ]

from (a,b) = F (P a) :×: F (P b)

A more complicated example shows how we can mix datatypes with a different
number of parameters, and partially instantiated datatypes:

data D α β γ = D β [ (α, Int) ] (RTree [γ ])
instance Generic (D α β γ) where

Rep (D α β γ) =
F (P 1) :×: F ([ ] :@: ‘[(,) :@: ‘[P 0,K Int]]) :×: F (RTree :@: ‘[[ ] :@: ‘[P 2]])

Pars (D α β γ) = ‘[α,β ,γ]

from (D a b c) = F (P a) :×: F (A b) :×: F (A c)

Even nested datatypes can be encoded, as our application supports the notion of
recursion with parameters instantiated to other applications:

data Perfect α = Perfect (Perfect (α,α)) | End α

instance Generic (Perfect α) where
Rep (Perfect α) = F (Perfect :@: ‘[(,) :@: ‘[P 0,P 0]]) :+: F (P 0)
Pars (Perfect α) = ‘[α]

from (Perfect x) = L (F (A x))
from (End x) = R (F (P x))

As we have seen, our generalisation of generic-deriving supports all the previous
datatypes, plus many new ones, involving any number of parameters of kind ?.

3.3 Mapping

While designing a new or improved generic programming library, defining the universe
and showing example datatype encodings is not enough; it is, in fact, easy to define
a simple and encompassing universe that is then found not to be suitable for defining
generic functions. As such, we show that our universe allows defining a generalised
map function, which maps n different functions over n datatype parameters.

Preliminaries As the generalised map takes n functions, one per datatype parameter,
we need a value-level counterpart to our type-level lists of parameters. We use a strongly
typed heterogenous list [4] for this purpose:

data HList (ρ :: [? ]) where
HNil :: HList ‘[ ]
HCons :: α → HList β → HList (α ‘: β )

We also need a way to pick the n-th element from such a list; we define a Lookup type
class for this purpose, as the function is defined by induction on the type-level natural
numbers. We use a Proxy to carry type information at the value-level (the index being
looked up):

data Proxy (σ :: κ) = Proxy
class Lookup (ρ :: [? ]) (ν :: Nat) where

lookup :: Proxy ν → HList ρ → ρ :!: ν

instance Lookup ρ Ze where
lookup (HCons f ) = f

instance (Lookup β ν)⇒ Lookup (α ‘: β ) (Su ν) where
lookup (HCons fs) = lookup (Proxy :: Proxy ν) fs

User-facing type class We are now ready to see the class that generalises Functor of
Section 2.3. This is a multi-parameter type class, taking a parameter σ for the unsat-
urated type we are mapping over, and a list of function types τ encoding the types of
the functions we will be mapping. As σ will often be ambiguous in the code for the
generic definition of map, we provide a version gmapP with an extra argument that
serves to identify the σ . A version without the proxy (gmap) is provided with a default
for convenient usage:

class GMap (σ :: κ) (τ :: [? ]) | τ → κ where
gmap :: HList τ → σ :$: Doms τ → σ :$: Codoms τ

gmap = gmapP (Proxy :: Proxy σ)

gmapP :: Proxy σ → HList τ → σ :$: Doms τ → σ :$: Codoms τ

Our generalised map, gmap, takes two arguments. The first is an HList of functions to
map. The second is the type σ applied to the domains of the functions we are mapping.
Its return type is again σ , but now applied to the codomains of the same functions. For
example, for lists, σ is [ ], and τ is ‘[α → β ]. In this case, gmap gets the expected type
HList ‘[α→ β ]→ [α ]→ [β ]. The functional dependency τ→ κ is essential to prevent
ambiguity when using gmap—and indeed the kind of σ is uniquely determined by the
(length of the) list τ . The type functions Doms and Codoms compute the domains and
codomains of the list of functions τ:

Doms (τ :: [? ]) :: [? ]
Doms ‘[ ] = ‘[ ]
Doms ((α → β ) ‘: τ) = α ‘: Doms τ

Codoms (τ :: [? ]) :: [? ]
Codoms ‘[ ] = ‘[ ]
Codoms ((α → β ) ‘: τ) = β ‘: Codoms τ

Mapping on the generic representation We have seen GMap, the user-facing type
class for the generalised map. Its counterpart for representation types is GMapR, which
operates on the interpretation of a representation:

class GMapR (υ :: Univ) (τ :: [? ]) where
gmapR :: HList τ → In υ (Doms τ)→ In υ (Codoms τ)

We now go through the instances of GMapR for each representation type. There is
nothing to map over in units, and sums and products are simply traversed through:

instance GMapR U τ where
gmapR U = U

instance (GMapR α τ,GMapR β τ)⇒ GMapR (α :+: β ) τ where
gmapR fs (L x) = L (gmapR fs x)
gmapR fs (R x) = R (gmapR fs x)

instance (GMapR α τ,GMapR β τ)⇒ GMapR (α :×: β ) τ where
gmapR fs (x :×: y) = gmapR fs x :×: gmapR fs y

Fields require more attention, so we define a separate type class GMapRF to handle
them:

instance (GMapRF υ τ)⇒ GMapR (F υ) τ where
gmapR fs (F x) = F (gmapRF fs x)

class GMapRF (υ :: Field) (τ :: [? ]) where
gmapRF :: HList τ → InField υ (Doms τ)→ InField υ (Codoms τ)

Constants, however, just like units, are returned unchanged:

instance GMapRF (K α) τ where
gmapRF (K x) = K x

For a parameter P ν , we need to lookup the right function to map over. We again use a
separate type class, GMapRP, and we traverse the input list of functions until we reach
the ν-th function. We thus require the list of functions τ to have its elements in the same
order as the datatype parameters:

instance (GMapRP ν τ)⇒ GMapRF (P ν) τ where
gmapRF fs (P x) = P (gmapRP (Proxy :: Proxy ν) fs x)

class GMapRP (ν :: Nat) (τ :: [? ]) where
gmapRP :: Proxy ν → HList τ → (Doms τ) :!: ν → (Codoms τ) :!: ν

instance GMapRP Ze ((α → β ) ‘: τ) where
gmapRP (HCons f ) x = f x

instance (GMapRP ν τ)⇒ GMapRP (Su ν) ((α → β ) ‘: τ) where
gmapRP (HCons fs) p = gmapRP (Proxy :: Proxy ν) fs p

Handling application The only representation type we still have to deal with is (:@:).
This is also the most challenging case. As a running example, consider the second
field of the D datatype of Section 3.2. D has three parameters, α , β , and γ , and the
second field of its only constructor has type [ (α, Int) ], represented as [ ] :@: ‘[(,) :@:
‘[P 0,K Int]]. In this situation, we intend to transform [ (α, Int) ] into [ (α ′, Int) ], having
available a function of type α→α ′. We do this by requiring the availability of gmap for
this particular list type; that is, we require an instance GMap [ ] ‘[(α, Int)→ (α ′, Int)].
Having such an instance, we can simply gmap over the argument. However, the func-
tions we map need to be adapted to this new argument. That is the task of AdaptFs,
which we explain below.

A type function MakeFs computes the types of the functions to be mapped inside the
applied type. For example, MakeFs ‘[(,) :@: ‘[P 0,K Int]] ‘[α → α ′,β → β ′,γ → γ ′] is
‘[(α, Int)→ (α ′, Int)]. We use proxies to fix otherwise ambiguous types when invoking
gmapP and adaptFs:

MakeFs (ρ :: [Field ]) (τ :: [? ]) :: [? ]
MakeFs ‘[ ] τ = ‘[ ]
MakeFs ((K α) ‘: ρ) τ = (α → α) ‘: MakeFs ρ τ

MakeFs ((P ν) ‘: ρ) τ = (τ :!: ν) ‘: MakeFs ρ τ

MakeFs ((σ :@: ω) ‘: ρ) τ =
(AppFields σ ω (Doms τ)→ AppFields σ ω (Codoms τ)) ‘: MakeFs ρ τ

instance (GMap σ (MakeFs ρ τ)
,AdaptFs ρ τ

,ExpandField ρ (Doms τ)∼ Doms (MakeFs ρ τ)
,ExpandField ρ (Codoms τ)∼ Codoms (MakeFs ρ τ)
)⇒ GMapRF (σ :@: ρ) τ where

gmapRF fs (A x) = A (gmapP p1 (adaptFs p2 fs) x)

where p1 = Proxy :: Proxy σ

p2 = Proxy :: Proxy ρ

This instance has four constraints, two of them being equality constraints, introduced
by the ∼ operator: a constraint of the form α ∼ β indicates that the type α must be
equal to the type β . The four constraints of this instance are: the ability to map over
the argument type, the requirement to rearrange the functions we map over, and two
coherence conditions on the behaviour of ExpandField and MakeFs over the list of
functions. The latter are always true for valid ρ and τ .

We are left with the task of adapting the functions to be mapped over. In our run-
ning example, we have an HList ‘[α → α ′,β → β ′,γ → γ ′], and we have to produce
an HList ‘[(α, Int)→ (α ′, Int)]. This is done by adaptFs, a method of the type class
AdaptFs which is implemented by induction on the list of fields to be mapped over:

class AdaptFs (ρ :: [Field ]) (τ :: [? ]) where
adaptFs :: Proxy ρ → HList τ → HList (MakeFs ρ τ)

For an empty list of an arguments, we return an empty list of functions. If the argument
is a constant, we ignore it and proceed to the next argument. For a parameter P ν , we
use the ν-th function of the original list of functions, and proceed to the next argument:

instance AdaptFs ‘[ ] τ where
adaptFs = HNil

instance (AdaptFs ρ τ)⇒ AdaptFs ((K α) ‘: ρ) τ where
adaptFs fs = HCons id (adaptFs (Proxy :: Proxy ρ) fs)

instance (AdaptFs ρ τ,Lookup τ ν)⇒ AdaptFs ((P ν) ‘: ρ) τ where
adaptFs fs = HCons (lookup p1 fs) (adaptFs p2 fs)

where p1 = Proxy :: Proxy ν

p2 = Proxy :: Proxy ρ

The most delicate case is, again, application. Back to our running example, this is where
we have to produce a function of type (α, Int)→ (α ′, Int). We do so by requiring an
instance GMap (,) ‘[α → α ′, Int→ Int], reusing MakeFs and AdaptFs in the process.
We also need to proceed recursively for the rest of the arguments. Again, we have the
same two coherence conditions on the behaviour of ExpandField and MakeFs over the
list of functions, and use proxies to fix ambiguous types:

instance (GMap σ (MakeFs ω τ)
,AdaptFs ω τ

,AdaptFs ρ τ

,ExpandField ω (Doms τ)∼ Doms (MakeFs ω τ)
,ExpandField ω (Codoms τ)∼ Codoms (MakeFs ω τ)
)⇒ AdaptFs ((σ :@: ω) ‘: ρ) τ where

adaptFs fs = HCons (gmapP p1 (adaptFs p2 fs)) (adaptFs p3 fs)
where p1 = Proxy :: Proxy σ

p2 = Proxy :: Proxy ω

p3 = Proxy :: Proxy ρ

The generic definition of the generalised map is thus complete, and ready to be used.

3.4 Example usage

Before we instantiate map to our example datatypes of Section 3.2, we first provide
a generic default [11] to make instantiation simpler. This default, for the GMap class,
will allow us to then give empty instances of the class, which will automatically use
the generic definition for the generalised map. The default version of gmapP converts
a value into its generic representation, applies gmapR, and then converts back to a user
datatype. This requires several constraints (which would all be inferred if the function
was defined at the top level). First, we introduce α and β as synonyms for the input and
output type, respectively, for mere convenience. In the case of the GMap [ ] ‘[γ → γ ′]
instance, for example, α is [γ ], and β is [γ ′ ]. We also require a Generic [γ ] instance (for
from), and a Generic [γ ′ ] instance (for to); these instances will coincide, and indeed we
also require that the representation Rep [γ ] be the same as Rep [γ ′ ] (which is the case).
Furthermore, the parameters of [γ ] have to coincide with the domains of ‘[γ → γ ′], and
the parameters of [γ ′ ] have to coincide with the codomains of the same list. Finally, we
require the ability to map generically over the representation of the input list:

default gmapP :: (α ∼ (σ :$: Doms τ)
,β ∼ (σ :$: Codoms τ)
,Generic α,Generic β

,Rep α ∼ Rep β

,Pars α ∼ Doms τ

,Pars β ∼ Codoms τ

,GMapR (Rep α) τ

)⇒ Proxy σ → HList τ → α → β

gmapP fs = to◦gmapR fs◦ from

With this default in place, we are ready to instantiate our example datatypes:

instance GMap [ ] ‘[α → α ′]
instance GMap RTree ‘[α → α ′]
instance GMap (,) ‘[α → α ′,β → β ′]
instance GMap D ‘[α → α ′,β → β ′,γ → γ ′]
instance GMap Perfect ‘[α → α ′]

Using the generic default, instantiation is kept simple and concise. We can verify that
our map works as expected in an example that should cover all the representation types:

x :: D Int Float Char
x = D 0.2 [(0,0),(1,1)] (RTree "p" [ ])
y :: D Int String Char
y = gmap (HCons (+1) (HCons show (HCons (const ’q’) HNil))) x

Indeed, y evaluates to D "0.2" [(1,0),(2,1)] (RTree "q" [ ]) as expected.

4 Limitations and future work

In this section we discuss the limitations of our new generic-deriving, and propose
future research directions.

4.1 Parameters of higher kinds

While our generalisation supports any number of parameters of kind ?, it is unable to
deal with parameters of higher kinds. Consider the following two datatypes representing
generalised trees:

data GTree1 φ α = GTree1 α (φ (GTree1 φ α))
data GTree2 α φ = GTree2 α (φ (GTree2 α φ))

The most general mapping function for GTree1 has the following type:

(α → β )→ (∀α β .(α → β )→ φ α → ψ β )→ GTree1 φ α → GTree1 ψ β

The generalisation of this paper provides a map of type (α → β )→ GTree1 φ α →
GTree1 φ β , therefore ignoring the φ parameter of kind ?→ ?. For GTree2, however,
our approach cannot even provide that simple map; we cannot give the GMap instance,
since the parameters of kind ? of interest come before a parameter of kind ?→ ?, and
the second parameter of GMap is a list of kind [? ] (so all arguments must have kind ?).

As such, we support generic abstraction only over the parameters of kind ? which
come after any parameters of other kinds. Lifting this restriction is not trivial. Recall
that the Pars type family has return kind [? ]. To generalise to multiple parameters, we
should make this return kind be a (promoted) heterogeneous list. This is currently not
possible, as heterogeneous lists are implemented as GADTs, which cannot be promoted.
Foregoing giving Pars the correct kind and working with nested tuples instead gives rise
to many kind ambiguities, which are cumbersome to solve. As such, we hope that the
promotion mechanism is extended to allow promotion of GADTs soon [15], and defer
generalising our approach to parameters of arbitrary kinds until then.

4.2 Integration with existing generic programming libraries

The introduction of our new generic-deriving raises the question of how to upgrade
from the old version. Our changes are not backwards compatible, but since the new
version encodes strictly more information than the previous one, we can provide a con-
version that automatically transforms the new representation into the old one, therefore
remaining compatible with old code [10]. The core of this conversion is a type-level
function Dn→D that converts the new representation into the old one. We show a proto-
type implementation here, subscripting the new generic-deriving codes with an n to
distinguish them from the old ones. Units, sums, and products are converted trivially.
Fields are handled by a separate function Dn→DF . Constants are converted trivially. For
a parameter, we return P if it is the last parameter of the datatype (the only one sup-
ported by the old version of generic-deriving), or a constant otherwise. We make
use of type-level if-then-else, equality on Nat, and length. Applications of types of kind
?→ ? are converted into compositions. For applications of types of higher arity, we first
apply the type to all its arguments but the last:

IfThenElse (α :: Bool) (β :: κ) (γ :: κ) :: κ

IfThenElse True β γ = β

IfThenElse False β γ = γ

(α :: Nat)≡Nat (β :: Nat) :: Bool
Ze≡Nat Ze = True
Su υ ≡Nat Ze = False
Ze≡Nat Su ν = False
Su υ ≡Nat Su ν = υ ≡Nat ν

Length (ρ :: [κ ]) :: Nat
Length ‘[ ] = Ze
Length (α ‘: ρ) = Su (Length ρ)

Dn→D (υ :: Univn) (ρ :: [? ]) :: Univ
Dn→D Un ρ = U
Dn→D (α :+n: β ) ρ = Dn→D α ρ :+: Dn→D β ρ

Dn→D (α :×n: β ) ρ = Dn→D α ρ :×: Dn→D β ρ

Dn→D (Fn α) ρ = Dn→DF α ρ

Dn→DF (υ :: Fieldn) (ρ :: [? ]) :: Univ
Dn→DF (Kn α) ρ = K α

Dn→DF (Pn ν) ρ = IfThenElse (ν ≡Nat Length ρ) P (K (ρ :!: ν))
Dn→DF (φ :@n: ‘[α]) ρ = φ :@: Dn→DF α ρ

Dn→DF (φ :@n: (α ‘: β ‘: γ)) ρ = Dn→DF ((AppFields φ ‘[α] ρ) :@n: (β ‘: γ)) ρ

The introduction of yet another generic programming library underscores the need
for a single mechanism for type reflection baked into the compiler, from which other
mechanisms, such as our new generic-deriving, or Data and Typeable [5], could then
be derived.

4.3 Parameter genericity vs. arity genericity

The approach described in this paper allows us to define generic functions that operate
over multiple datatype parameters. This is not the same as generic functions that operate
at diverse arities [14]. Consider the following generic functions:

map1
n :: HList ‘[α1

1 , . . . ,α
1
n ]→ φ (α1

1 . . .α
1
n )

map2
n :: HList ‘[α1

1 → α2
1 , . . . ,α

1
n → α2

n ]→ φ (α1
1 . . .α

1
n )→ φ (α2

1 . . .α
2
n )

map3
n :: HList ‘[α1

1 → α2
1 → α3

1 , . . . ,α
1
n → α2

n → α3
n ]

→ φ (α1
1 . . .α

1
n )→ φ (α2

1 . . .α
2
n )→ φ (α3

1 . . .α
3
n )

mapm
n :: HList ‘[α1

1 → . . .→ αm
1 , . . . ,α

1
n → . . .→ αm

n ]

→ φ (α1
1 . . .α

1
n )→ . . .→ φ (αm

1 . . .αm
n )

The function map1
n, or repeat, creates a φ -structure given elements for the parameter po-

sitions. The function map2
n, equivalent to the generic gmap of Section 3.3, is the function

that maps over a φ -structure, taking one function per parameter. The function map3
n, or

zipWith, is the function that takes two φ -structures and zips them when their shapes are
compatible, using the provided functions to zip the parameters. The function mapm

n is
the generalisation of the previous three, in the arity-generic sense. Our approach allows
defining each of map1

n, map2
n, map3

n, etc., individually, as separate generic functions. It
does not allow defining mapm

n ; that generalisation is described by Weirich and Casingh-

ino [14], in the dependently-typed programming language Agda. It remains to see how
to transfer the concept of arity-genericity to Haskell.

5 Conclusion

In this paper we have seen how we can use the promotion mechanism together with kind
polymorphism to encode a generic representation of datatypes that supports abstraction
over multiple parameters (of kind ?). This enables a whole new class of generic func-
tionality: we have shown a generalised map, but also folding, traversing, and zipping,
for example, are now possible. We plan to implement support for the new generic-de-
riving in GHC soon, so that users can take advantage of the new functionality without
needing to write their own Generic instances.

